D03PCF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

D03PCF integrates a system of linear or nonlinear parabolic partial differential equations (PDEs) in one space variable. The spatial discretisation is performed using finite differences, and the method of lines is employed to reduce the PDEs to a system of ordinary differential equations (ODEs). The resulting system is solved using a backward differentiation formula method.

2 Specification

```
 SUBROUTINE D03PCF(NPDE, M, TS, TOUT, PDEDEF, BNDARY, U, NPTS, X,
1                  ACC, W, NW, IW, NIW, ITASK, ITRACE, IND, IFAIL)
 INTEGER           NPDE, M, NPTS, NW, IW(NIW), NIW, ITASK, ITRACE,
1                  IND, IFAIL
 real              TS, TOUT, U(NPDE,NPTS), X(NPTS), ACC, W(NW)
 EXTERNAL          PDEDEF, BNDARY
```

3 Description

D03PCF integrates the system of parabolic equations:

$$\sum_{j=1}^{\text{NPDE}} P_{i,j}\frac{\partial U_j}{\partial t} + Q_i = x^{-m}\frac{\partial}{\partial x}(x^m R_i), \quad i = 1,2,\ldots,\text{NPDE}, \quad a \le x \le b, \;\; t \ge t_0, \tag{1}$$

where $P_{i,j}$, Q_i and R_i depend on x, t, U, U_x and the vector U is the set of solution values

$$U(x,t) = [U_1(x,t),\ldots,U_{\text{NPDE}}(x,t)]^T, \tag{2}$$

and the vector U_x is its partial derivative with respect to x. Note that $P_{i,j}$, Q_i and R_i must not depend on $\frac{\partial U}{\partial t}$.

The integration in time is from t_0 to t_{out}, over the space interval $a \le x \le b$, where $a = x_1$ and $b = x_{\text{NPTS}}$ are the leftmost and rightmost points of a user-defined mesh $x_1, x_2, \ldots, x_{\text{NPTS}}$. The co-ordinate system in space is defined by the value of m; $m = 0$ for Cartesian co-ordinates, $m = 1$ for cylindrical polar co-ordinates and $m = 2$ for spherical polar co-ordinates. The mesh should be chosen in accordance with the expected behaviour of the solution.

The system is defined by the functions $P_{i,j}$, Q_i and R_i which must be specified in a subroutine PDEDEF supplied by the user.

The initial values of the functions $U(x,t)$ must be given at $t = t_0$. The functions R_i, for i = 1,2,...,NPDE, which may be thought of as fluxes, are also used in the definition of the boundary conditions for each equation. The boundary conditions must have the form

$$\beta_i(x,t)R_i(x,t,U,U_x) = \gamma_i(x,t,U,U_x), \quad i = 1,2,\ldots,\text{NPDE}, \tag{3}$$

where $x = a$ or $x = b$.

The boundary conditions must be specified in a subroutine BNDARY provided by the user.

The problem is subject to the following restrictions:

(i) $t_0 < t_{out}$, so that integration is in the forward direction;
(ii) $P_{i,j}$, Q_i and the flux R_i must not depend on any time derivatives;
(iii) The evaluation of the functions $P_{i,j}$, Q_i and R_i is done at the mid-points of the mesh intervals by calling the routine PDEDEF for each mid-point in turn. Any discontinuities in these functions **must** therefore be at one or more of the mesh points $x_1, x_2, \ldots, x_{\text{NPTS}}$;

(iv) At least one of the functions $P_{i,j}$ must be non-zero so that there is a time derivative present in the problem; and

(v) If $m > 0$ and $x_1 = 0.0$, which is the left boundary point, then it must be ensured that the PDE solution is bounded at this point. This can be done by either specifying the solution at $x = 0.0$ or by specifying a zero flux there, that is $\beta_i = 1.0$ and $\gamma_i = 0.0$. See also Section 8 below.

The parabolic equations are approximated by a system of ODEs in time for the values of U_i at mesh points. For simple problems in Cartesian co-ordinates, this system is obtained by replacing the space derivatives by the usual central, three-point finite-difference formula. However, for polar and spherical problems, or problems with nonlinear coefficients, the space derivatives are replaced by a modified three-point formula which maintains second-order accuracy. In total there are NPDE × NPTS ODEs in the time direction. This system is then integrated forwards in time using a backward differentiation formula method.

4 References

[1] Berzins M (1990) Developments in the NAG Library software for parabolic equations *Scientific Software Systems* (ed J C Mason and M G Cox) Chapman and Hall 59–72

[2] Berzins M, Dew P M and Furzeland R M (1989) Developing software for time-dependent problems using the method of lines and differential-algebraic integrators *Appl. Numer. Math.* **5** 375–397

[3] Skeel R D and Berzins M (1990) A method for the spatial discretization of parabolic equations in one space variable *SIAM J. Sci. Statist. Comput.* **11 (1)** 1–32

[4] Dew P M and Walsh J (1981) A set of library routines for solving parabolic equations in one space variable *ACM Trans. Math. Software* **7** 295–314

5 Parameters

1: NPDE — INTEGER *Input*

On entry: the number of PDEs in the system to be solved.

Constraint: NPDE ≥ 1.

2: M — INTEGER *Input*

On entry: the co-ordinate system used, m:

M = 0
: indicates Cartesian co-ordinates,

M = 1
: indicates cylindrical polar co-ordinates,

M = 2
: indicates spherical polar co-ordinates.

Constraint: $0 \leq \text{M} \leq 2$.

3: TS — ***real*** *Input/Output*

On entry: the initial value of the independent variable t.

On exit: the value of t corresponding to the solution values in U. Normally TS = TOUT.

Constraint: TS < TOUT.

4: TOUT — ***real*** *Input*

On entry: the final value of t to which the integration is to be carried out.

5: PDEDEF — SUBROUTINE, supplied by the user. *External Procedure*

PDEDEF must compute the functions $P_{i,j}$, Q_i and R_i which define the system of PDEs. PDEDEF is called approximately midway between each pair of mesh points in turn by D03PCF.

Its specification is:

```
SUBROUTINE PDEDEF(NPDE, T, X, U, UX, P, Q, R, IRES)
INTEGER           NPDE, IRES
real              T, X, U(NPDE), UX(NPDE), P(NPDE,NPDE), Q(NPDE),
1                 R(NPDE)
```

1: NPDE — INTEGER *Input*
On entry: the number of PDEs in the system.

2: T — ***real*** *Input*
On entry: the current value of the independent variable t.

3: X — ***real*** *Input*
On entry: the current value of the space variable x.

4: U(NPDE) — ***real*** array *Input*
On entry: U(i) contains the value of the component $U_i(x,t)$, for $i = 1,2,\ldots,$NPDE.

5: UX(NPDE) — ***real*** array *Input*
On entry: UX(i) contains the value of the component $\frac{\partial U_i(x,t)}{\partial x}$, for $i = 1,2,\ldots,$NPDE.

6: P(NPDE,NPDE) — ***real*** array *Output*
On exit: P(i,j) must be set to the value of $P_{i,j}(x,t,U,U_x)$, for $i,j = 1,2,\ldots,$NPDE.

7: Q(NPDE) — ***real*** array *Output*
On exit: $Q(i)$ must be set to the value of $Q_i(x,t,U,U_x)$, for $i,j = 1,2,\ldots,$NPDE.

8: R(NPDE) — ***real*** array *Output*
On exit: $R(i)$ must be set to the value of $R_i(x,t,U,U_x)$, for $i = 1,2,\ldots,$NPDE.

9: IRES — INTEGER *Input/Output*
On entry: set to -1 or 1.

On exit: should usually remain unchanged. However, the user may set IRES to force the integration routine to take certain actions as described below:

IRES = 2
indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator set to IFAIL = 6.

IRES = 3
indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. The user may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If the user consecutively sets IRES = 3, then D03PCF returns to the calling (sub)program with the error indicator set to IFAIL = 4.

PDEDEF must be declared as EXTERNAL in the (sub)program from which D03PCF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

6: BNDARY — SUBROUTINE, supplied by the user. *External Procedure*

BNDARY must compute the functions β_i and γ_i which define the boundary conditions as in equation (3).

Its specification is:

```
SUBROUTINE BNDARY(NPDE, T, U, UX, IBND, BETA, GAMMA, IRES)
INTEGER           NPDE, IBND, IRES
real              T, U(NPDE), UX(NPDE), BETA(NPDE), GAMMA(NPDE)
```

1: NPDE — INTEGER *Input*

On entry: the number of PDEs in the system.

2: T — ***real*** *Input*

On entry: the current value of the independent variable t.

3: U(NPDE) — ***real*** array *Input*

On entry: U(i) contains the value of the component $U_i(x,t)$ at the boundary specified by IBND, for $i = 1, 2, \ldots$,NPDE.

4: UX(NPDE) — ***real*** array *Input*

On entry: UX(i) contains the value of the component $\frac{\partial U_i(x,t)}{\partial x}$ at the boundary specified by IBND, for $i = 1, 2, \ldots$,NPDE.

5: IBND — INTEGER *Input*

On entry: IBND determines the position of the boundary conditions. If IBND = 0, then BNDARY must set up the coefficients of the left-hand boundary $x = a$. Any other value of IBND indicates that BNDARY must set up the coefficients of the right-hand boundary, $x = b$.

6: BETA(NPDE) — ***real*** array *Output*

On exit: BETA(i) must be set to the value of $\beta_i(x,t)$ at the boundary specified by IBND, for $i = 1, 2, \ldots$,NPDE.

7: GAMMA(NPDE) — ***real*** array *Output*

On exit: GAMMA(i) must be set to the value of $\gamma_i(x,t,U,U_x)$ at the boundary specified by IBND, for $i = 1, 2, \ldots$,NPDE.

8: IRES — INTEGER *Input/Output*

On entry: set to -1 or 1.

On exit: should usually remain unchanged. However, the user may set IRES to force the integration routine to take certain actions as described below:

IRES = 2 indicates to the integrator that control should be passed back immediately to the calling (sub)program with the error indicator set to IFAIL = 6.

IRES = 3 indicates to the integrator that the current time step should be abandoned and a smaller time step used instead. The user may wish to set IRES = 3 when a physically meaningless input or output value has been generated. If the user consecutively sets IRES = 3, then D03PCF returns to the calling (sub)program with the error indicator set to IFAIL = 4.

BNDARY must be declared as EXTERNAL in the (sub)program from which D03PCF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

7: U(NPDE,NPTS) — ***real*** array *Input/Output*

On entry: the initial values of $U(x,t)$ at t = TS and the mesh points X(j), for j = 1,2,...,NPTS.

On exit: U(i,j) will contain the computed solution at t = TS.

8: NPTS — INTEGER *Input*

On entry: the number of mesh points in the interval $[a,b]$.

Constraint: NPTS ≥ 3.

9: X(NPTS) — ***real*** array *Input*

On entry: the mesh points in the spatial direction. X(1) must specify the left-hand boundary, a, and X(NPTS) must specify the right-hand boundary, b.

Constraint: X(1) < X(2) < ... < X(NPTS).

10: ACC — ***real*** *Input*

On entry: a positive quantity for controlling the local error estimate in the time integration. If E(i, j) is the estimated error for U_i at the jth mesh point, the error test is:

$$|\mathrm{E}(i,j)| = \mathrm{ACC} \times (1.0 + |\mathrm{U}(i,j)|).$$

Constraint: ACC > 0.0.

11: W(NW) — ***real*** array *Workspace*

12: NW — INTEGER *Input*

On entry: the dimension of the array W as declared in the (sub)program from which D03PCF is called.

Constraint: NW ≥ (10+6×NPDE) × NPDE × NPTS + (21+3×NPDE) × NPDE + 7 × NPTS + 54.

13: IW(NIW) — INTEGER array *Output*

On exit: the following components of the array IW concern the efficiency of the integration.

IW(1) contains the number of steps taken in time.

IW(2) contains the number of residual evaluations of the resulting ODE system used. One such evaluation involves evaluating the PDE functions at all the mesh points, as well as one evaluation of the functions in each of the boundary conditions.

IW(3) contains the number of Jacobian evaluations performed by the time integrator.

IW(4) contains the order of the last backward differentiation formula method used.

IW(5) contains the number of Newton iterations performed by the time integrator. Each iteration involves an ODE residual evaluation followed by a back-substitution using the LU decomposition of the Jacobian matrix.

The rest of the array is used as workspace.

14: NIW — INTEGER *Input*

On entry: the dimension of the array IW as declared in the (sub)program from which D03PCF is called.

Constraint: NIW = NPDE × NPTS + 24.

15: ITASK — INTEGER *Input*

On entry: specifies the task to be performed by the ODE integrator. The permitted values of ITASK and their meanings are detailed below:

ITASK = 1
: normal computation of output values U at t = TOUT.

ITASK = 2
: one step and return.

ITASK = 3
: stop at first internal integration point at or beyond t = TOUT.

Constraint: 1 ≤ ITASK ≤ 3.

16: ITRACE — INTEGER *Input*

On entry: the level of trace information required from D03PCF and the underlying ODE solver. ITRACE may take the value -1, 0, 1, 2, or 3. If ITRACE < -1, then -1 is assumed and similarly if ITRACE > 3, then 3 is assumed. If ITRACE $= -1$, no output is generated. If ITRACE $= 0$, only warning messages from the PDE solver are printed on the current error message unit (see X04AAF). If ITRACE > 0, then output from the underlying ODE solver is printed on the current advisory message unit (see X04ABF). This output contains details of Jacobian entries, the nonlinear iteration and the time integration during the computation of the ODE system. The advisory messages are given in greater detail as ITRACE increases. Users are advised to set ITRACE $= 0$, unless they are experienced with the subchapter D02M–N of the NAG Fortran Library.

17: IND — INTEGER *Input/Output*

On entry: IND must be set to 0 or 1.

IND $= 0$
starts or restarts the integration in time.

IND $= 1$
continues the integration after an earlier exit from the routine. In this case, only the parameters TOUT and IFAIL should be reset between calls to D03PCF.

Constraint: $0 \leq \text{IND} \leq 1$.

On exit: IND $= 1$.

18: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL $= 1$

On entry, TOUT $\leq$ TS,
or (TOUT $-$ TS) is too small,
or ITASK $\neq 1$, 2 or 3,
or M $\neq 0$, 1 or 2,
or M > 0 and X(1) < 0.0,
or X(i), for $i = 1, 2, \ldots,$NPTS are not ordered,
or NPTS < 3,
or NPDE < 1,
or ACC ≤ 0.0,
or IND $\neq 0$ or 1,
or NW is too small,
or NIW is too small,
or D03PCF called initially with IND $= 1$.

IFAIL $= 2$

The underlying ODE solver cannot make any further progress, across the integration range from the current point $t =$ TS with the supplied value of ACC. The components of U contain the computed values at the current point $t =$ TS.

IFAIL = 3

In the underlying ODE solver, there were repeated errors or corrector convergence test failures on an attempted step, before completing the requested task. The problem may have a singularity or ACC is too small for the integration to continue. Integration was successful as far as t = TS.

IFAIL = 4

In setting up the ODE system, the internal initialisation routine was unable to initialise the derivative of the ODE system. This could be due to the fact that IRES was repeatedly set to 3 in the user-supplied subroutines PDEDEF or BNDARY, when the residual in the underlying ODE solver was being evaluated.

IFAIL = 5

In solving the ODE system, a singular Jacobian has been encountered. The user should check his problem formulation.

IFAIL = 6

When evaluating the residual in solving the ODE system, IRES was set to 2 in at least one of the user-supplied subroutines PDEDEF or BNDARY. Integration was successful as far as t = TS.

IFAIL = 7

The value of ACC is so small that the routine is unable to start the integration in time.

IFAIL = 8

In one of the user-supplied routines, PDEDEF or BNDARY, IRES was set to an invalid value.

IFAIL = 9

A serious error has occurred in an internal call to D02NNF. Check problem specification and all parameters and array dimensions. Setting ITRACE = 1 may provide more information. If the problem persists, contact NAG.

IFAIL = 10

The required task has been completed, but it is estimated that a small change in ACC is unlikely to produce any change in the computed solution. (Only applies when the user is not operating in one step mode, that is when ITASK $\neq$ 2.)

IFAIL = 11

An error occurred during Jacobian formulation of the ODE system (a more detailed error description may be directed to the current advisory message unit).

IFAIL = 12

Not applicable.

IFAIL = 13

Not applicable.

IFAIL = 14

The flux function R_i was detected as depending on time derivatives, which is not permissible.

7 Accuracy

The routine controls the accuracy of the integration in the time direction but not the accuracy of the approximation in space. The spatial accuracy depends on both the number of mesh points and on their distribution in space. In the time integration only the local error over a single step is controlled and so the accuracy over a number of steps cannot be guaranteed. The user should therefore test the effect of varying the accuracy parameter, ACC.

8 Further Comments

The routine is designed to solve parabolic systems (possibly including some elliptic equations) with second-order derivatives in space. The parameter specification allows the user to include equations with only first-order derivatives in the space direction but there is no guarantee that the method of integration will be satisfactory for such systems. The position and nature of the boundary conditions in particular are critical in defining a stable problem. It may be advisable in such cases to reduce the whole system to first-order and to use the Keller box scheme routine D03PEF.

The time taken by the routine depends on the complexity of the parabolic system and on the accuracy requested.

9 Example

We use the example given in Dew and Walsh [4] which consists of an elliptic-parabolic pair of PDEs. The problem was originally derived from a single third-order in space PDE. The elliptic equation is

$$\frac{1}{r}\frac{\partial}{\partial r}\left(r^2\frac{\partial U_1}{\partial r}\right) = 4\alpha\left(U_2 + r\frac{\partial U_2}{\partial r}\right)$$

and the parabolic equation is

$$(1-r^2)\frac{\partial U_2}{\partial t} = \frac{1}{r}\frac{\partial}{\partial r}\left(r\left(\frac{\partial U_2}{\partial r} - U_2U_1\right)\right)$$

where $(r,t) \in [0,1] \times [0,1]$. The boundary conditions are given by

$$U_1 = \frac{\partial U_2}{\partial r} = 0 \text{ at } r = 0,$$

and

$$\frac{\partial}{\partial r}(rU_1) = 0 \text{ and } U_2 = 0 \text{ at } r = 1.$$

The first of these boundary conditions implies that the flux term in the second PDE, $\left(\frac{\partial U_2}{\partial r} - U_2U_1\right)$, is zero at $r = 0$.

The initial conditions at $t = 0$ are given by

$$U_1 = 2\alpha r \text{ and } U_2 = 1.0, \text{ for } r \in [0,1].$$

The value $\alpha = 1$ was used in the problem definition. A mesh of 20 points was used with a circular mesh spacing to cluster the points towards the right-hand side of the spatial interval, $r = 1$.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     D03PCF Example Program Text
*     Mark 19 Revised. NAG Copyright 1999.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NPDE, NPTS, INTPTS, ITYPE, NEQN, NIW, NWK, NW
      PARAMETER        (NPDE=2,NPTS=20,INTPTS=6,ITYPE=1,NEQN=NPDE*NPTS,
     +                 NIW=NEQN+24,NWK=(10+6*NPDE)*NEQN,
     +                 NW=NWK+(21+3*NPDE)*NPDE+7*NPTS+54)
*     .. Scalars in Common ..
      real             ALPHA
```

```
*     .. Local Scalars ..
      real             ACC, HX, PI, PIBY2, TOUT, TS
      INTEGER          I, IFAIL, IND, IT, ITASK, ITRACE, M
*     .. Local Arrays ..
      real             U(NPDE,NPTS), UOUT(NPDE,INTPTS,ITYPE), W(NW),
     +                 X(NPTS), XOUT(INTPTS)
      INTEGER          IW(NIW)
*     .. External Functions ..
      real             X01AAF
      EXTERNAL         X01AAF
*     .. External Subroutines ..
      EXTERNAL         BNDARY, D03PCF, D03PZF, PDEDEF, UINIT
*     .. Intrinsic Functions ..
      INTRINSIC        SIN
*     .. Common blocks ..
      COMMON           /VBLE/ALPHA
*     .. Data statements ..
      DATA             XOUT(1)/0.0e+0/, XOUT(2)/0.40e+0/,
     +                 XOUT(3)/0.6e+0/, XOUT(4)/0.8e+0/,
     +                 XOUT(5)/0.9e+0/, XOUT(6)/1.0e+0/
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D03PCF Example Program Results'
      ACC = 1.0e-3
      M = 1
      ITRACE = 0
      ALPHA = 1.0e0
      IND = 0
      ITASK = 1
*
*     Set spatial mesh points
*
      PIBY2 = 0.5e0*X01AAF(PI)
      HX = PIBY2/(NPTS-1)
      X(1) = 0.0e0
      X(NPTS) = 1.0e0
      DO 20 I = 2, NPTS - 1
         X(I) = SIN(HX*(I-1))
   20 CONTINUE
*
*     Set initial conditions
*
      TS = 0.0e0
      TOUT = 0.1e-4
      WRITE (NOUT,99999) ACC, ALPHA
      WRITE (NOUT,99998) (XOUT(I),I=1,6)
*
*     Set the initial values
*
      CALL UINIT(U,X,NPTS)
      DO 40 IT = 1, 5
         IFAIL = -1
         TOUT = 10.0e0*TOUT
*
         CALL D03PCF(NPDE,M,TS,TOUT,PDEDEF,BNDARY,U,NPTS,X,ACC,W,NW,IW,
     +               NIW,ITASK,ITRACE,IND,IFAIL)
*
*        Interpolate at required spatial points
```

```
*
         CALL D03PZF(NPDE,M,U,NPTS,X,XOUT,INTPTS,ITYPE,UOUT,IFAIL)
         WRITE (NOUT,99996) TOUT, (UOUT(1,I,1),I=1,INTPTS)
         WRITE (NOUT,99995) (UOUT(2,I,1),I=1,INTPTS)
   40 CONTINUE
*
*     Print integration statistics
*
      WRITE (NOUT,99997) IW(1), IW(2), IW(3), IW(5)
      STOP
*
99999 FORMAT (//' Accuracy requirement  = ',e12.5,/' Parameter ALPHA =',
     +       '       ',e12.3,/)
99998 FORMAT ('   T  /  X  ',6F8.4,/)
99997 FORMAT (' Number of integration steps in time                  ',
     +       I4,/' Number of residual evaluations of resulting ODE sys',
     +       'tem',I4,/' Number of Jacobian evaluations               ',
     +       '         ',I4,/' Number of iterations of nonlinear solve',
     +       'r              ',I4,/)
99996 FORMAT (1X,F6.4,' U(1)',6F8.4)
99995 FORMAT (8X,'U(2)',6F8.4,/)
      END
*
      SUBROUTINE UINIT(U,X,NPTS)
*     Routine for PDE initial conditon
*     .. Scalar Arguments ..
      INTEGER          NPTS
*     .. Array Arguments ..
      real             U(2,NPTS), X(NPTS)
*     .. Scalars in Common ..
      real             ALPHA
*     .. Local Scalars ..
      INTEGER          I
*     .. Common blocks ..
      COMMON           /VBLE/ALPHA
*     .. Executable Statements ..
      DO 20 I = 1, NPTS
         U(1,I) = 2.0e0*ALPHA*X(I)
         U(2,I) = 1.0e0
   20 CONTINUE
      RETURN
      END
*
      SUBROUTINE PDEDEF(NPDE,T,X,U,DUDX,P,Q,R,IRES)
*     .. Scalar Arguments ..
      real             T, X
      INTEGER          IRES, NPDE
*     .. Array Arguments ..
      real             DUDX(NPDE), P(NPDE,NPDE), Q(NPDE), R(NPDE),
     +                 U(NPDE)
*     .. Scalars in Common ..
      real             ALPHA
*     .. Common blocks ..
      COMMON           /VBLE/ALPHA
*     .. Executable Statements ..
      Q(1) = 4.0e0*ALPHA*(U(2)+X*DUDX(2))
      Q(2) = 0.0e+0
      R(1) = X*DUDX(1)
```

```
      R(2) = DUDX(2) - U(1)*U(2)
      P(1,1) = 0.0e+0
      P(1,2) = 0.0e0
      P(2,1) = 0.0e+0
      P(2,2) = 1.0e0 - X*X
      RETURN
      END
*
      SUBROUTINE BNDARY(NPDE,T,U,UX,IBND,BETA,GAMMA,IRES)
*     .. Scalar Arguments ..
      real              T
      INTEGER           IBND, IRES, NPDE
*     .. Array Arguments ..
      real              BETA(NPDE), GAMMA(NPDE), U(NPDE), UX(NPDE)
*     .. Executable Statements ..
      IF (IBND.EQ.0) THEN
         BETA(1) = 0.0e+0
         BETA(2) = 1.0e+0
         GAMMA(1) = U(1)
         GAMMA(2) = -U(1)*U(2)
      ELSE
         BETA(1) = 1.0e0
         BETA(2) = 0.0e+0
         GAMMA(1) = -U(1)
         GAMMA(2) = U(2)
      END IF
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D03PCF Example Program Results


 Accuracy requirement  =  0.10000E-02
 Parameter ALPHA =          0.100E+01

   T  /  X    0.0000  0.4000  0.6000  0.8000  0.9000  1.0000

 0.0001 U(1)  0.0000  0.8008  1.1988  1.5990  1.7958  1.8485
        U(2)  0.9997  0.9995  0.9994  0.9988  0.9663  0.0000

 0.0010 U(1)  0.0000  0.7982  1.1940  1.5841  1.7179  1.6734
        U(2)  0.9969  0.9952  0.9937  0.9484  0.6385  0.0000

 0.0100 U(1)  0.0000  0.7676  1.1239  1.3547  1.3635  1.2830
        U(2)  0.9627  0.9495  0.8754  0.5537  0.2908  0.0000

 0.1000 U(1)  0.0000  0.3908  0.5007  0.5297  0.5120  0.4744
        U(2)  0.5468  0.4299  0.2995  0.1479  0.0724  0.0000

 1.0000 U(1)  0.0000  0.0007  0.0008  0.0008  0.0008  0.0007
        U(2)  0.0010  0.0007  0.0005  0.0002  0.0001  0.0000
```

```
Number of integration steps in time                     78
Number of residual evaluations of resulting ODE system 378
Number of Jacobian evaluations                          25
Number of iterations of nonlinear solver               190
```
